

Customize EXCEL Output

Hui Liu, Jiangsu Hengrui Pharmaceuticals Co.,Ltd., Guang Zhou

ABSTRACT

Exporting data from SAS® data sets to Microsoft EXCEL spreadsheets has been a commonly requested task for SAS programmers, but the conventional methods seldom have the option to generate highly formatted EXCEL reports. ODS EXCEL Destination is a powerful and convenient tool to make this possible. This paper will give a step by step demonstration with examples on how to use many of the options and sub-options of ODS EXCEL Destination as well as PROC REPORT procedure to customize EXCEL output, and will highlight its effectiveness over ODS EXCELXP Tagset.

INTRODUCTION

Traditionally, we have several ways to convert SAS® data sets to EXCEL spreadsheets, such as PROC EXPORT, DDE, LIBNAME statement. However, most of them can only export SAS® data into EXCEL spreadsheets, but couldn't generate customized reports or graphs. ODS EXCELXP Tagset had the capability to format the output data and worksheets, but will not create a native EXCEL file. ODS EXCEL Destination is available to produce customized native EXCEL (XLSX) files since the release of SAS® 9.4 (TS1M3), which can write output to all SAS® Procedures, such as PROC PRINT, PROC REPORT, PROC SGPLOT, PROC SGRENDER, PROC MEANS, etc. This paper is aimed at beginner to intermediate level programmers with experience creating tables with the REPORT procedure.

This paper will focus on the following features and benefits of ODS Excel Destination:

- Create an index tab of all worksheets
- Update the orientation of printed page
- Specify repeated rows to print
- Define tab color
- Frozen headers or columns
- Add automatic filter
- Insert title and footnote with hyperlink
- Customize header text and column width
- Add hyperlink to a special cell
- Add comment to a special cell
- Customize font color and background color
- Create multiple tabs

CUSTOMIZE EXCEL OUTPUT

ODS EXCEL Destination has several options and sub-options to customize the printing features and other common EXCEL settings, by using appropriate options, we can generate a pretty-looking EXCEL output without opening the EXCEL workbook.

ODS EXCEL OPTIONS

Output 1 shows an example of some commonly used ODS EXCEL Options:

The file= defines the destination of the excel file and file name;

The sheet_name= specifies the name for the worksheet;

- The protect_worksheet = protects the worksheet by making it read-only;
- The index= creates a worksheet that contains an index with hyperlinks of all worksheets;
- The orientation= orients the printed page as either portrait or landscape;
- The row_repeat= repeats all the row headers for each page;
- The tab_color= specifies the color for the worksheet;
- The start_at= specifies a starting cell for the report;
- The frozen_headers= specifies whether to freeze the headers;
- The autofilter= turns on filtering for specified columns in the worksheet;
- The embedded_titles= specifies whether title should appear in the worksheet;
- The embedded_footnotes= specifies whether footnote should appear in the worksheet;
- The hidden_columns= specifies the columns to hide.

```
ods listing close;
ods excel file="C:\temp\pharmasug\sample_output.xlsx"
options(sheet_name="Demographic" protect_worksheet="on" index="on"
orientation='LANDSCAPE' row_repeat='header' start_at="2,2"
tab_color="Yellow" frozen_headers="yes" frozen_rowheaders="1"
autofilter="1-8" embedded_titles="yes" embedded_footnotes="yes"
hidden_columns='5');
```

Output 1. ODS EXCEL Options

The screenshot shows an Excel spreadsheet with three tabs: 'Index', 'Demographic', and 'Stock Prices'. The 'Demographic' tab is highlighted in yellow. Below the tabs, a data table is shown with the following structure:

Subject ID	Treatment Group	Age	Race	Height at Baseline	Weight at Baseline	BMI at Baseline
1011001	Treatment A	24	AMERICAN INDIAN OR ALASKA NATIVE	172	80	27.04
1011002	Treatment A	60	ASIAN	160	45	17.58

Annotations in the image explain the ODS options:

- index=on**: Points to the 'Index' tab.
- tab_color=yellow**: Points to the yellow 'Demographic' tab.
- start_at="2,2"**: Points to the start of the data table (row 2, column 2).
- hidden_columns="5"**: Points to column E, which is hidden.
- row_repeat='header'**: Points to the header row of the data table, indicating it will repeat on each page.
- autofilter="1-8"**: Points to the header row, indicating that columns 1 through 8 will have filters applied.
- frozen_headers="yes"**: Points to the header row, indicating that the headers are frozen.

Display 1. ODS EXCEL Options

OPTIONS FOR TITLE AND FOOTNOTE

Output 2 shows an example of specifying title and footnote:

- The j= specifies justification of title is center, left or right;
- The height= specifies the size of the font for titles;
- The italic specifies that the title text is in italics;
- The color specifies the color of title;

The `underlin=` specifies whether the subsequent text is underlined;

The `link=` adds a hyperlink for the subsequent text.

```
proc report data=dm nowd missing style(header)={just=c color=black
backgroundcolor=lightgray fontsize=10pt} split='^';
  title1 j=1 height=8pt italic color=black "Protocol Numer:SHR-XXXX" ;
  title2 j=c color=blue underlin=1 link="#"Index"!a1" "Demographic
Information" ;
  title3 j=c color=black "Safety Analysis Set";
  footnote1 j=1 color=black "Safety analysis set refer to subjects who take
at least one dose of study drug.";
```

Output 2. Define Title and Footnote

Subject ID	Treatment Group	Age	Race	Height at Baseline	Weight at Baseline	BMI at Baseline
1011040	Treatment B	42	WHITE	188	82	23.20
1011041	Treatment B	38	ASIAN	168	68	24.09
1011042	Treatment B	19	ASIAN	170	55	19.03

Display 2. Header, Title and Footnote Layout

DEFINE HEADER TEXT AND COLUMN WIDTH

Output 3 shows an example of specifying header text and column width:

The `backgroundcolor=` specify the background color of the header;

The “text” in define statement specifies the text to display in header, if variable label is the exact text we want to display in header, no need to specify “text” here;

The `cellwidth=` specify the column width.

Apart from define statement, ODS sub-option `ABSOLUTE_COLUMN_WIDTH='number-list'` can also be used to specify column width.

```
proc report data=dm nowd missing style(header)={just=c color=black
backgroundcolor=lightgray fontsize=10pt} split='^';
  define subjid /display 'Subject ID' style=[cellwidth=0.6in];
  define trt01p /display 'Treatment Group' style=[cellwidth=0.8in
background=$color_fmt.];
```

Output 3. Specify Header Text and Column Width

Subject ID	Treatment Group	Age	Race	Height at Baseline	Weight at Baseline	BMI at Baseline
------------	-----------------	-----	------	--------------------	--------------------	-----------------

Display 3. Header Layout

ADD HYPERLINK TO A SPECIFIC CELL

Output 4 shows the use of a separate COMPUTE statement for each report-item variable that you want to add hyperlink. Conditional variables (if race=) that are used to define the hyperlink of the desired rows, which may not be the same variable as the COMPUTE report-item variable, but must be listed before the report-item variable. In the CALL DEFINE statement below, _COL_ indicates that a hyperlink will be added for the COMPUTE report-item variable if the condition met, URL= specifies the link.

```
compute race;
  if race='AMERICAN INDIAN OR ALASKA NATIVE' then call
define(_col_, "style", "style=[color=blue
url='https://minorityhealth.hhs.gov/omh/browse.aspx?lvl=3&lvlid=62']");
endcomp;
```

Output 4. Add Hyperlink

Age	Race	Height at Baseline	Weight at Baseline
42	WHITE	188	82
38	ASIAN	168	68
19	ASIAN	170	55
24	AMERICAN INDIAN OR ALASKA NATIVE	172	80
60	ASIAN		45
45	AMERICAN		72

https://minorityhealth.hhs.gov/omh/browse.aspx?lvl=3&lvlid=62 - 单击一次可跟踪超链接。单击并接任不放手可选择此单元格。

Display 4. Hyperlink for a Special Cell

ADD COMMENT

Output 5 uses a separate COMPUTE statement to add a comment to a specific cell, FLYOVER= to specify the text of the comment.

```
compute sex;
  if sex='F' then call define(_col_, "style", "style=[color=red
flyover='Refer to Female']");
  if sex='M' then call define(_col_, "style", "style=[color=green
flyover='Refer to Male']");
endcomp;
```

Output 5. Add Comment

Subject ID	Treatment Group	Age	Gender	Race
1011016	Treatment B	42	M	WHITE
1011017	Treatment B	38	M	Refer to Male
1011018	Treatment B	19	F	

Flyover="Refer to Male"

Display 5. Comment for a Cell

CUSTOMIZE FONT COLOR & BACKGROUND COLOR

There are 2 ways to customize the cell attribute, Output 6 uses a separate COMPUTE statement to specify the font color and background color; Output 7 uses predefined SAS® format to specify cell background color.

```

compute bmi;
  if bmi>26 or bmi<19 then call
define(_col_, "style", "style=[color=green backgroundColor=lightblue]");
endcomp;

```

Output 6. Compute Statement to Define Font Color & Background Color

Subject ID	Treatment Group	Age	Race	Height at Baseline	Weight at Baseline	BMI at Baseline
1011001	Treatment A	24	AMERICAN INDIAN OR ALASKA NATIVE	172	80	27.04
1011002	Treatment A	60	ASIAN	160	45	17.58
1011003	Treatment A	45	BLACK OR AFRICAN AMERICAN	180	72	22.22
1011004	Treatment B	31	NATIVE HAWAIIAN OR OTHER PACIFIC ISLANDER	191	70	19.19

color=green, backgroundColor=lightblue in COMPUTE statement

Display 6. Compute Statement to Define Font Color & Background Color

```

proc format;
  value $color_fmt
    "Treatment A" = "pink"
    "Treatment B" = "lightyellow";
run;

define trt01p /display 'Treatment Group' style(column)=[cellwidth=0.8in
background=$color_fmt.];

```

Output 7. SAS Format to Define Font Color & Background Color

Subject ID	Treatment Group	Age	Race	Height at Baseline	Weight at Baseline	BMI at Baseline
1011001	Treatment A	24	AMERICAN INDIAN OR ALASKA NATIVE	172	80	27.04
1011002	Treatment A	60	ASIAN	160	45	17.58
1011003	Treatment A	45	BLACK OR AFRICAN AMERICAN	180	72	22.22
1011004	Treatment B	31	NATIVE HAWAIIAN OR OTHER PACIFIC ISLANDER	191	70	19.19

background=\$color_fmt. in define statement


Display 7. SAS Format to Define Font Color & Background Color

ADDING MULTIPLE TABS

To add multiple tabs, ODS EXCEL options can be used again after the first report is generated, but unnecessary to specify FILE= since the same workbook will be created with different tabs. SHEET_NAME= is used to specify the new tab name, NOGTITLE prints the title outside of the graph borders. Please note the firstly defined options will be retained for the following tabs unless a new value is assigned. Output 8 add FOOTNOTE option to avoid the footnote in the first tab to be displayed.

```
ods excel options (sheet_name="Stock Prices" start_at="5,4"
frozen_rowheaders="off" tab_color="Pink" embedded_titles="yes" ) nogtitle;
ods graphics / reset width=6in height=3in;
footnote;
proc sgplot data=sashelp.stocks;
  title color=blue underlin=1 link="#"Index"!a2" "Stock Prices 1986-
2005";
  series x=date y=close / group=stock;
run;
```

Output 8. Adding Multiple Tabs



Display 8. Adding a New Tab

DIFFERENCE BETWEEN ODS EXCEL DESTINATION AND ODS EXCELXP TAGSET

ODS EXCELXP Tagset almost has the same functionalities as ODS EXCEL Destination, but there are still some differences between them.

FILE EXTENSION

The ODS EXCELXP Tagset generates files in the Microsoft Office XML spreadsheet format, the file extension is XLS or XLM; whereas the ODS Excel Destination generates files in native EXCEL format (XLSX), the native EXCEL format is much smaller than the XML spreadsheet format. Output 9 and Output 10 input the same options and same dataset, but Display 9 shows the file size of the two are significantly different. This benefit is particularly useful when we deliver large files via email.



Besides, each time we open the XLS file generated by ODS EXCELXP Tagset, a warning dialog box like Display 10 will pop up, since the actual format we created via ODS EXCEL XP Tagset is not native EXCEL format.

```
ods listing close;
ods excel file="C:\temp\pharmasug\test-excel.xlsx"
options(sheet_name="Class" orientation='LANDSCAPE' row_repeat='header'
start_at="2,2"
tab_color="Yellow" frozen_headers="yes" frozen_rowheaders="1"
autofilter="1-6") ;
proc print data=sashelp.class;run;
ods _all_ close;
```


Output 9. ODS EXCEL Destination Code

```
ods listing close;
ods tagsets.excelxp path="C:\temp\pharmasug\" file="test-tagset.xls"
options(sheet_name="Class" orientation='LANDSCAPE' row_repeat='header'
start_at="2,2"
tab_color="Yellow" frozen_headers="yes" frozen_rowheaders="1"
autofilter="1-6") ;
proc print data=sashelp.class;run;
ods _all_ close;
```

Output 10. ODS EXCELXP Tagset Code

 test-excel.xlsx	8 KB
 test-tagset.xls	73 KB

Display 9. File Size of Two Output



Display 10. Warning Dialog Box

DEFAULT STYLE

The default style of ODS EXCEL Destination is SAS EXCEL style, the layout is soft blue on white (Display 11). While the ODS EXCELXP Tagset use the SAS default style, which is gray on gray (Display 12), the appearance of ODS EXCEL Destination is much good-looking than ODS EXCELXP Tagset. But the pros of ODS EXCELXP Tagset is its flexibility to modify the tagset template as necessary.

	A	B	C	D	E	F
1	Class Information					
2						
3	Obs	Name	Sex	Age	Height	Weight
4	1	Alfred	M	14	69.0	112.5
5	2	Alice	F	13	56.5	84.0
6	3	Barbara	F	13	65.3	98.0
7	4	Carol	F	14	62.8	102.5
8	5	Henry	M	14	63.5	102.5
9	6	James	M	12	57.3	83.0
10	7	Jane	F	12	59.8	84.5
11	8	Janet	F	15	62.5	112.5
12	9	Jeffrey	M	13	62.5	84.0
13	10	John	M	12	59.0	99.5
14	11	Joyce	F	11	51.3	50.5
15	12	Judy	F	14	64.3	90.0
16	13	Louise	F	12	56.3	77.0
17	14	Mary	F	15	66.5	112.0
18	15	Philip	M	16	72.0	150.0
19	16	Robert	M	12	64.8	128.0
20	17	Ronald	M	15	67.0	133.0
21	18	Thomas	M	11	57.5	85.0
22	19	William	M	15	66.5	112.0

Display 11. ODS EXCEL Destination Layout


	A	B	C	D	E	F
1		<i>Class Information</i>				
2						
3		Obs	Name	Sex	Age	Height
4		1	Alfred	M	14	69
5		2	Alice	F	13	56.5
6		3	Barbara	F	13	65.3
7		4	Carol	F	14	62.8
8		5	Henry	M	14	63.5
9		6	James	M	12	57.3
10		7	Jane	F	12	59.8
11		8	Janet	F	15	62.5
12		9	Jeffrey	M	13	62.5
13		10	John	M	12	59
14		11	Joyce	F	11	51.3
15		12	Judy	F	14	64.3
16		13	Louise	F	12	56.3
17		14	Mary	F	15	66.5
18		15	Philip	M	16	72
19		16	Robert	M	12	64.8
20		17	Ronald	M	15	67
21		18	Thomas	M	11	57.5
22		19	William	M	15	66.5

Display 12. ODS EXCELXP Tagset Layout

ODS EXCEL DESTINATION AND ODS EXCELXP TAGSET OPTIONS

Some new functionalities are added in ODS EXCEL Destination, but does not exist in ODS EXCELXP Tagset. Such as START_AT which enables us to specify the start row and column of the report in ODS EXCEL Destination, but no such function for ODS EXCELXP Tagset. Same case for PROTECT_WORKSHEET to make the EXCEL file read only.

Another new sub-option for ODS EXCEL Destination is FLOW=. If we use WIDTH= style attribute to define the column width, ODS EXCEL Destination uses an algorithm to determine where text in a cell should wrap for the best presentation. When the text wraps within a cell, carriage return/line feed (CRLF) characters are inserted where the wrapping occurs. However, the ODS EXCELXP Tagset will not insert a carriage return/line feed when wrap text. To prevent the hard return, FLOW="TABLES" sub-option should be used.



Display 13. Wrap Text for ODS EXCEL Destination and ODS EXCELXP Tagset

GRAPH

Another benefit of ODS EXCEL Destination is the capability to inserting graph to workbook while ODS EXCELXP Tagset can't.

CONCLUSION

ODS EXCEL Destination is a very powerful and convenient tool to make highly formatted EXCEL reports, whether to provide data issue report to Data Management team, or to provide special EXCEL format report requested by medical writer, even to generate an EXCEL format table, listing or figure, such as patient profiles, each tab can be used to map one subject's information, ODS EXCEL Destination is the first preference choice. Apart from ODS options and sub-options, we can also use PROC REPORT procedure to further customize the attribute of columns, rows, and cells.

REFERENCES

Benjamin, William E., Jr. 2018. "The ODS Excel Destination: Assorted Tips and Techniques to Simplify Writing SAS® Data to Excel Workbooks." *WUSS*, 153-2018.

SAS Help Center. "REPORT Procedure." Accessed May 24, 2021.
https://documentation.sas.com/doc/en/pgmsascdc/9.4_3.5/proc/n1b1be5822k8nnn1s1ucv8fvpq3d.htm.

SAS Help Center. "ODS EXCEL Statement." Accessed May 5, 2021.
https://documentation.sas.com/doc/en/pgmsascdc/9.4_3.5/odsug/p09n5pw9ol0897n1qe04zeur27rv.htm.

RECOMMENDED READING

- *SAS Output Delivery System: User's Guide*
- *Base SAS Procedures Guide*

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Hui Liu
Jiangsu Hengrui Pharmaceuticals Co.,Ltd.
Hui.liu.hl3@hengrui.com
www.hengrui.com

SAS® and all other SAS® Institute Inc. product or service names are registered trademarks or trademarks of SAS® Institute Inc. in the USA and other countries. ® indicates USA registration.

Any brand and product names are trademarks of their respective companies.